Podmínky úspěšně odevzdané semestrální práce z dynamiky
1) Semestrální práce musí být kompletně vypracována v libovolném matematickém programu (Matlab, wxMaxima, Scilab, Octave, ...) a převedena do pdf.
2) Semestrální práce musí obsahovat cvičícím podepsané zadání.
3) Každý krok řešení bude komentovaný (proč, odkud, ...).
4) Semestrální práce bude odeslána ve dvou souborech na adresu cvičícího. Ve formátu Jméno_Příjmení_verze.pdf a Jméno_Příjmení_verze.xxx pro script (přípona je závislá na použitém softwaru). V poli předmět e-mailu vyplňte Dynamika - semestrální práce.
Pro tento mnou vytvořený dokument tedy Michal_Sivčák_01.wxmx a Michal_Sivčák_01.pdf.
Poslední termín odevzdání je ke konci prvního týdne zkouškového období. Po tomto termínu již žádné práce nebudou přijímány (ani opravené)!!

pozn. S.P. by měla obsahovat úvod (co budu řešit a proč, odkud čerpám, na co navazuji), vlastní řešení s grafy s popisky os a závěr, kde budou stručně shrnuty dosažené výsledky.

Dynamika - semestrální práce (vzor)

Vypracoval: Michal Sivčák
Cvičení: Sobota 0:00
odevzdáno: leden 3000

Figure 1: Podepsané zadání S.P.



1 Zadání

V klasické mechanice je pohyb kyvadla podrobně rozebrán. V základních kurzech se obvykle uchylujeme k řešení pro malé úhly rozkmitu. Zkusme se tedy podívat, co se s kyvadlem, resp. periodou jeho kyvu stane při obecném počátečním úhlu. Úkolem této práce je tedy naleznout závislost mezi počátečním úhlem a dobou kyvu kyvadla s délkou závěsu 1m. Omezíme se na interval 1°-179°. Pro 0° je pozice kyvadla stabilní a nepohybuje se a pro 180° je poloha kyladla labilní (nicméně se opět nepohybuje - při nekonečně malém vychýlení z labilní polohy by došlo k pohybu, teoreticky by ale trval tento pohyb nekonečně dlouho).

2 Řešení

Nejprve vyčistíme paměť programu Maxima.
pozn. všechny úhly v Maximě musí být v radiánech, proto všude používáme přepočet α[rad]=α[°]·pi/180.

```
(%i1)   kill(all)$
```

Při řešení vyjdeme ze zákona o zachování mechanické energie a položíme počáteční energii rovnu mechanické energii v obecné poloze.

(%i1) rovnice:K0+V0=K+V;

(rovnice) $V0+K0=V+K$

Kde V0 je potenciální energie na počátku

(%i2) V0:m·g·(−R·cos(alpha_0));

(V0) $-R\cos(\alpha_0)\, g\, m$

K0 je kinetická energie na počátku, vzhledem k tomu, že v0=0, je i K0=0

(%i3) K0:0;

(K0) 0

K je kinetická energie v obecné poloze

(%i4) K:1/2·m·v^2;

(K) $\frac{m\, v^2}{2}$

a nakonec V je potenciální energie v obecné poloze

(%i5) V:m·g·(−R·cos(alpha));

(V) $-R\cos(\alpha)\, g\, m$

Po dosazení do rovnice (1) dostaneme

(%i6) rovnice:ev(rovnice);

(rovnice) $-R\cos(\alpha_0)\, g\, m=\frac{m\, v^2}{2}-R\cos(\alpha)\, g\, m$

Odtud vyjádříme rychlost jako funkci úhlu α

(%i7) reseni:solve([rovnice], [v]);

(reseni) $[v=-\sqrt{2}\sqrt{R\cos(\alpha)\, g-R\cos(\alpha_0)\, g}, v=\sqrt{2}\sqrt{R\cos(\alpha)\, g-R\cos(\alpha_0)\, g}]$

Budeme-li předpokládat, že se kyvadlo nachází ve výchozí poloze v pravé části a kladný úhel α bude kótován kladně proti směru hodinových ručiček, bude se v prvním kmitu pohybovat hmotný bod směrem doleva po směru hodinových ručiček. Z výše uvedeného vyplývá, že znaménko plus platí pro pohyb proti směru a znaménko minus po směru hodinových ručiček. Jelikož se nám jedná o dobu kyvu, stačí řešit pouze jednu fázi pohybu a volíme tedy znaménko minus.

(%i8)
```
v:rhs(reseni[1]);
```

(v) $$-\sqrt{2}\sqrt{R\cos(\alpha)\,g-R\cos(\alpha_0)\,g}$$

Podívejme se na graf funkce v(α) při počátečním úhlu 179° (tento kyv bude nejdelší a informace o délce tohoto kyvu se nám bude hodit později)

(%i9)
```
wxplot2d(ev(v,R=1,g=9.81,alpha_0=179·%pi/180),
    [alpha,−179·%pi/180,179·%pi/180],
    [xlabel, "úhel α (rad)"], [ylabel, "rychlost bodu (m/s)"]);
```

(%t9)



(%o9)

Upravíme výraz (8) pomocí vztahu v/R=diff(α,t)

(%i10)
```
drov:'diff(alpha,t)=v/R;
```

(drov) $$\frac{d}{d\,t}\,\alpha=-\frac{\sqrt{2}\sqrt{R\cos(\alpha)\,g-R\cos(\alpha_0)\,g}}{R}$$

Výše uvedená diferenciální rovnice je nelineární a musíme ji řešit numericky. Dosadíme do pravé strany konkrétní hodnoty.

(%i11)
```
prstr:ev(rhs(drov),R=1,g=9.81,alpha_0=179·%pi/180);
```

(prstr) $$-\sqrt{2}\sqrt{9.81\cos(\alpha)-9.81\cos\left(\frac{179\,\pi}{180}\right)}$$

Pro výpočet použijeme Runge-Kuttovu metodu (vzhledem k vlastnostem metody a diferenciální rovnice je potřeba nepatrně zmenšit počáteční úhel, tj. vynásobíme jej "skoro jedničkou"). Pro nastavení metody použijene délku simulace 10s s krokem 0.001s a počáteční podmínkou α0=0.9999·179°.

(%i12)
```
vysledek:rk(prstr,alpha,(0.9999·179)·%pi/180,[t,0,10,0.001])$
```

Výstupem metody jsou hodnoty času a příslušné souřadnice (úhlu) v jednotlivých krocích výpočtu. Např. pro desátý krok výpočtu vypadá výsledek následovně:

(%i13)
```
vysledek[10];
```

(%o13) [0.009000000000000001 , 3.123726374555555]

Všechny výsledky si můžeme zobrazit v grafu

(%i14)
```
wxplot2d ([discrete, vysledek],[xlabel, "čas (s)"], [ylabel, "α (rad)"]);
```

(%t14)



(%o14)

Je zřejmé, že pro určitý čas T se výchylka nemění (protože počítáme pouze s pohybem doleva). Tento čas musíme nalézt, tzn. že musíme v cyklu projít všechny výsledky RK metody dokud nebudou po sobě jdoucí hodnoty výhylky shodné. Z grafu je patrné, že maximální délka kyvu bude menší než 4s. Stačí tedy procházet prvních 4000 výsledků (čas/krok=4/0.001=4000).

```
(%i15) for i:1 thru 4000 do
         if vysledek[i][2] = vysledek[i+1][2] then (T:vysledek[i][1], i:4000);
(%o15) done
```

```
(%i16) T;
(%o16) 3.853
```

Vymažeme dosud získané výsledky

```
(%i17) kill(i,T,vysledek,prstr);
(%o17) done
```

Takto jsme nalezli dobu periody pouze pro jeden případ kdy phi_0=179°. Nyní stačí výše uvedený výpočet projít v krocích od 0° do 179° po jednom stupni a máme hotovo.

```
(%i18) for k:1 thru 179 step 1 do
         (
         prstr:ev(rhs(drov),R=1,g=9.81,alpha_0=ev(k·%pi/180)),
         vysledek:rk(ev(prstr),alpha,ev(0.9999·k·%pi/180),[t,0,4,0.001]),
           for i:1 thru 4000 do
             if vysledek[i][2] = vysledek[i+1][2] then
                         (T[k]:vysledek[i][1], i:4000),
         kill(i,alpha,vysledek,t)
         );
ARRSTORE: use_fast_arrays=false; allocate a new property hash table for |$t|
(%o18) done
```

Vytvoříme seznam hodnot pro vykreslení grafu

```
(%i19) graf:makelist([i,T[i]],i,1,179)$
```

Graf hledané závislosti na intervalu 0°-180° vypadá následovně:

(%i20)
```
wxplot2d([discrete, graf],[grid2d,true],[xlabel, "počáteční úhel (°)"],
    [ylabel, "doba kyvu (s)"],[y,0,4])$
```

(%t20)



Pro zajímavost si zobrazme detail pro malé úhly 0°-20°

(%i21)
```
wxplot2d([discrete, graf],[grid2d,true],[xlabel, "počáteční úhel (°)"],
    [ylabel, "doba kyvu (s)"],[x,0,20],[y,0.99,1.01])$
```

(%t21)



3 Závěr

Pomocí zákona o zachování energie byla odvozena pohybová rovnice ve tvaru nelineární diferenciální rovnice prvního řádu. Tato rovnice byla pomocí cyklu numericky řešena Runge-Kuttovou metodou pro různé počáteční úhly vychýlení kyvadla při nulové počáteční rychlosti. Zároveň byla pro každý výsledek nalezena doba kyvu T taková, kdy rychlost kyvadla byla rovna nule.
Z grafu (t20) je patrné, že doba kyvu je nelineární a limitně se blíží k nekonečnu pro počáteční úhel blížící se 180°. Z grafu (t21) vyplývá, že se pro malé úhly doba kyvu prakticky nemění a do 15° je změna menší než 0.5%.